

SEM Programming Language

SEM SeSAM Simulator Programming Language

SYNOPSIS

SEM

Introduction to and overview of the Programming Language SEM offering the specification of a simulation and behaviour model for state-based agents used for the SeSAM simulator environment.

Author: Stefan Bosse

| | |
|--------------------------|----|
| SYNOPSIS | 1 |
| DESCRIPTION | 1 |
| AGENTS | 1 |
| DEFINITIONS | 3 |
| DATA TYPES | 4 |
| VALUES | 4 |
| EXPRESSIONS | 5 |
| PATHS | 6 |
| LISTS, ITERATORS, ARRAYS | 7 |
| SEQUENTIAL COMPOSITION | 8 |
| BRANCHES | 9 |
| LOOPS | 10 |
| SHAPES | 11 |
| EXAMPLES | 12 |
| VERSION | 18 |

DESCRIPTION

The activity-based functional SEM programming language provides statements for describing the behaviour of state-based agents by using activities and transitions connecting and enabling activities. Activities and functions provide generic functional statements to perform computation and actions. The SEM programming language is a text level design interface for the Multi-Agent System (MAS) simulation environment SeSAM (Klügel et al.), which provides only a GUI. It is closely related to the SeSAM agent and world model, but is extended with some convenient functions and statements, easing the design of complex MAS.

AGENTS

An agent belongs to a particular agent class defining body variables, activities, and transitions between activities, summarized in Table 1.

1.2 8/4/14

Tab. 1. Agent Class Definition

| SEM Statement | Description |
|---|---|
| <pre>agent ac(shapedef) = definitions ↵ variables import activities transitions end;</pre> | Definition of an agent class consisting of variable definitions, import of feature classes, activities, and transitions. The shape definition parameter list is optional. |
| <pre>world wc = definitions ↵ variables import activities transitions end;</pre> | Definition of a world class consisting of variable definitions, import of feature classes, activities, and transitions. |
| <pre>resource rc(shapedef) = definitions ↵ variables import end;</pre> | Definition of a resources class consisting of variable definitions and import of feature classes. The shape definition parameter list is optional. |
| <pre>def x = expr end; def mutable x = expr end; def private mutable x = expr end; def x:T = expr end;</pre> | Definition of global read only, global mutable, and private data storage objects of data type DT derived from the initial value or specified with an explicit type declaration. |
| <pre>activity a = [statements] before; [statements]; [statements] after; transitions end;</pre> | Definition of an agent activity representing the state of the agent. Statements of the activity are executed in sequential order before, immediately, or after an activity was activated by a transition. |
| <pre>transition (A_i, A_j, cond_{ij}); transition (A_i, A_j);</pre> | Definition of agent state transitions (conditional depending on evaluation of a boolean expression and unconditional). |

| SEM Statement | Description |
|---|---|
| <pre>feature fc = definitions end; open fc;</pre> | <p>Definition of a feature class including variable and function definitions. Within agent, world or feature classes other feature classes can be imported by using the open statement.</p> |

SEM

DEFINITIONS

Value (variable), function, and type definitions are summarized in Tab. 2.

Tab. 2. Definition of types, values, and functions

| Statement | Description |
|---|---|
| <pre>type e = {X,Y,Z,..};</pre> | <p>Definition of an enumeration (sum) type with symbol type elements X,Y,Z...</p> |
| <pre>type r = (x=T1, y=T2,..); ⇒ fun r: T1*T2*.. → r;</pre> | <p>Definition of a record structure type consisting of elements x,y.. with specified data types DT. Each record type definition introduces an automatic definition of a type constructor function with same name, which can be used in expressions to create a record type value.</p> |
| <pre>def v = expr end; def mutable v = expr end; def v:DT = expr end;</pre> | <p>Definition of an immutable (constant) and mutable value (variable). The type is either derived from the expression (not always possible) or specified explicitly by an additional type declaration.</p> |
| <pre>def f = fun x,y,.. → expr end; def f = fun x:T1, y:T2,.. → expr end;</pre> | <p>Definition of a (named) function with function parameters x,y. The datatype of the parameters can be specified explicitly by an additional type declaration. Type inference of function parameters is limited.</p> |

| Statement | Description |
|--|--|
| <code>fun F : T1*T2*.. → RT ;</code> | Declaration of a function type interface (supported only for builtin primitive functions). |

SEM DATA TYPES

Predefined ordinal data types `DT` are summarized in Table 7.

Tab. 3. Ordinal data types `DT`

| APL Statement | Type <code>DT</code> | Description |
|--|-----------------------|--|
| <code>integer</code> | <code>INT</code> | Signed integer |
| <code>double</code> | <code>REAL</code> | Floating point type |
| <code>string</code> | <code>STRING</code> | Text string consisting of characters |
| <code>boolean</code> | <code>BOOL</code> | Boolean type (with value set {true, false}). |
| <code>void</code> | - | Empty type |
| <code>'a list</code> <code>DT list</code> | <code>LIST</code> | Polymorphic and concrete list type |
| <code>'a iterator</code> <code>DT iterator</code> | <code>ITERATOR</code> | Polymorphic and concrete iterator type |
| <code>activity</code> | <code>ADT</code> | Activity type |
| <code>color</code> | <code>ADT</code> | Simulation object color type |
| <code>objclass</code> | <code>ADT</code> | Agent class type |
| <code>objinstance</code> | <code>ADT</code> | Agent class instance type |
| <code>position</code> | <code>ADT</code> | Simulation object position type |
| <code>simobject</code> | <code>ADT</code> | Simulation object type |
| <code>spatialinfo</code> | <code>ADT</code> | Spatial info type |

VALUES

Values of different data types DT can appear in expressions and assignments, summarized in Table 9.

Tab. 4. Values and data types DT

| Value | Type | Description |
|---------------------------------|--------|---|
| -2, -1, 0, 1, 2, 3, 4, ... | INT | Signed integer number (decimal format) |
| -2.41, ..., 2.41, ... | REAL | Floating type number (decimal format) |
| "abc" | STRING | String (character array) |
| true, false | BOOL | Boolean value |
| {v1, v2, ...} { } | LIST | List of values and empty list |
| [v1, v2, ...] | ARRAY | Array of values |
| T(v1, v2, ...) T(e1:v1, ...) | RECORD | Record value constructor function for record type T with optional labels specifying the record element. |
| ^class | CLASS | Instance of an agent class |

SEM

EXPRESSIONS

Expressions are used in assignments, branches, function applications, and loops. There are arithmetic, relational, and boolean operations.

Tab. 5. Arithmetic, relational, and boolean/logical (bitwise) operators with applicable data types

| Operator | Type | Description |
|------------|-----------|--|
| +, -, *, / | INT, REAL | Addition, Subtraction (Negation), Multiplication, Division |
| % | INT | Modulo division |

| Operator | Type | Description |
|---|---|---|
| < <= > >= = <> | INT, REAL | Lower than, lower equal than, greater than, greater equal than, equal, not equal. |
| and, or, , not | BOOL | Boolean operators |
| (x::DT) | DT' -> DT | Type casting of variables only |
| float(x) round(x) trunc(x) floor(x) ceiling(x) int(x) char(x) | DT -> REAL REAL -> INT REAL -> INT REAL -> INT REAL -> INT DT -> INT DT -> CHAR | Type conversion (applicable to expressions, values, variables) |

All operators of an expression must have the same type. Explicit type conversion can be used to convert a native data type to the expression type.

Function application is provided by using the function name and an argument list, which can be empty (Def. 1). Arguments containing expressions are evaluated before function application. Function applications can be embedded in expressions.

Def. 1. Function application (f) and procedure execution (p) with arguments, w/o arguments

```

dst := f(arg1, arg2, ...);
dst := f();
expr(f(...))
p(arg1, arg2, ...);
p();

```

PATHS

Access of objects (variables) from other agents is performed by using paths, summarized in Tab. 6. A path selector requires a root variable pointing to a valid simulation object. Path selectors can be used in expressions and on the LHS of an assignment.

Tab. 6. Path selectors

| Statement | Description |
|---|---|
| <pre>def x:simobject = null end; x->class->vref</pre> | <p>A variable <i>x</i> containing a valid simulation object reference is the root element of the path selector, which resolve a variable <i>vref</i> of an agent belonging to the specified <i>class</i>.</p> |

SEM

LISTS, ITERATORS, ARRAYS

Lists are dynamic data structures, which can be modified at run-time, summarized in Table 11. Iterators are derived from lists and are used by iterative functions provided by SeSAM. Some function except lists directly, other require conversion to an iterator object.

There is no array support in SeSAM. For this reason, arrays are emulated using hash tables. The SEM language provides limited array support.

Tab. 7. Summary of list definitions and their usage

| Statement | Data Type DT | Description |
|--------------------------------------|--------------|---|
| <pre>def L:DT list = {} end;</pre> | * LIST | Definition of a list variable initialized with an empty list. Initialization with empty lists requires the type declaration of the variable, otherwise type inference is used to determine the list type. |
| <pre>L := {v1, v2, ..};</pre> | * LIST | Assignment of a new list to a list variable. |

Tab. 8. List and set type definitions (TYPE: structure type)

| Statement | Data Type DT | Description |
|--|----------------------|---|
| <code>{v1, v2, v3, ..}</code> | * LIST | List of values |
| <code>L.[expr]</code> <code>L.[head]</code> <code>L.[tail]</code> | * LIST | Selection of a list element using an index expression (head and tail are keywords - the tail index is evaluated at run-time). |
| <code>L.[expr] := e;</code> <code>L.[head] := e;</code> <code>L.[tail] := e;</code> | * LIST | Modifies a selected list element using an index expression. |
| <code>L@{v1, v2, ..}</code> | * LIST | Concatenation of lists |
| <code>e :: L</code> | * LIST | Add an element to the top of the list L. |
| <code>L ::+ e;</code> <code>L +:: e;</code> | * LIST | Append new element <i>e</i> at the end or before the head of the list L. |
| <code>x ::- L;</code> <code>x -:: L;</code> | * LIST | Remove last or first element from list L and assign the removed element to the variable <i>x</i> . |
| <code>fun AsIterator:</code> <code>'a list -></code> <code>'a iterator;</code> <code>fun AsList:</code> <code>'a iterator -></code> <code>'a list;</code> | * LIST * ITERATOR | Functions for conversion between iterators and lists. |

SEQUENTIAL COMPOSITION

Statements in functions and activities are executed strict sequentially. Statements S_1 ; S_2 ; S_3 ; ... separated by a semicolon must be wrapped with a block statement, summarized in Tab. 8

Tab. 9. Sequential composition using blocks

| Statement | Description |
|---|--|
| <pre>[statement1; statement2; .. statementn];</pre> | Generic block which can be used in functions or nested activity statements. Statements grouped in a block are executed in a strict sequential order. |

SEM

BRANCHES

There are different branch statements available. They pass the program flow to an alternative statement or a block of statements depending on values. Branches can appear within block statements in functions or activities. There are functional counterparts to the procedural statements which may appear in expressions (conditional expressions).

Tab. 10. Procedural and functional branch statements

| Statement | Kind | Description |
|---|------------------------------|--|
| <pre>if cond then statement1 else statement0;</pre> | Boolean Branch Procedural | Depending on the result of the boolean expression <i>cond</i> a branch occurs either to <i>statement1</i> (<i>expr</i> =true) or to the optional alternative <i>statement0</i> (<i>cond</i> =false). |
| <pre>if cond then expr1 else expr2</pre> | Boolean Branch Functional | Depending on the result of the boolean expression <i>cond</i> either <i>expr1</i> (<i>cond</i> =true) or the required <i>expr0</i> (<i>cond</i> =false) is evaluated and its value is returned. |

| Statement | Kind | Description |
|--|----------------------------------|--|
| <pre> case <i>expr</i> of <i>v1</i> => <i>stmt1</i>; <i>v2</i> => <i>stmt2</i>; end; </pre> | Multi-value Branch Procedural | Different constant values are compared with the result of the expression <i>expr</i> and the respective statements are selected on successful matching. There is no default <code>else</code> case (matching all other values)! |
| <pre> case <i>expr</i> of <i>v1</i> => <i>expr1</i> <i>v2</i> => <i>expr2</i> </pre> | Multi-value Branch Functional | Different constant values are compared with the result of the expression <i>expr</i> and the respective expressions are evaluated on successful matching. There is no default <code>else</code> case (matching all other values)! A functional case branch must be complete and must contain all possible cases (so it is limited to enumeration types). |

LOOPS

There are different loop statements available. Each loop repeats the execution of the loop body as long as a boolean condition is satisfied. A counting loop iterates a list of values, either specified explicitly by a set/list or implicitly by a range set constructor.

Tab. 11. Loop statements

| Statement | Kind | Description |
|---|------------------|---|
| <pre> for i = a to downto b do statement done; for i in x .. for i in {v1,..} .. </pre> | Counting Loop | <p>The for-loop executes the loop body <i>statements</i> for each element in the iterator list, either a range of values or a set/list (variable <i>x</i>) of values. The loop iterator variable <i>i</i> holds the current value taken from the list.</p> <p>The range includes the limiting values <i>a</i> and <i>b</i>.</p> |
| <pre> while expr do statement done; </pre> | Conditional Loop | <p>The while-loop executes the loop body as long as the boolean expression <i>expr</i> is true. The test of the boolean expression is performed before each loop iteration.</p> |

SEM

SHAPES

Agent and resources classes are related to spatially located geometric objects, which can be (initially) specified as a parameter list of a class.

Tab. 12. Shape parameter

| Statement | Kind | Description |
|---------------------------|----------|--|
| <code>color: color</code> | COLOR | Shape colors: grey, lightgrey, yellow, green, blue, red, orange. |
| <code>shape: shape</code> | GEOMETRY | Shape geometry: rectangle, square, circle, ellipse. |

| Statement | Kind | Description |
|--|----------|---|
| <code>size: {width, height}</code> | GEOMETRY | Shape size specifying the width and height of the shape (double value). |
| <code>center: {x0, y0}</code> | POSITION | Relative shape center point (double value). |
| <code>fill: bool</code> | COLOR | Shape fill attribute (true/false). |

SEM

EXAMPLES

Ex. 1. Resource class definitions

```
resource link(color:grey,
             shape:rectangle,
             size:{2.0,2.0},center:{1.0,1.0}) =
  def mutable Node = Position(0,0) end;
  def mutable Link = NORTH end;
end;
resource connection(color:blue,
                   shape:rectangle,
                   size:{2.0,2.0},
                   center:{1.0,1.0}) =
  def mutable Node1:simobject = null end;
  def mutable Node2:simobject = null end;
end;
```

Ex. 2. Type definitions

```
type Stats = (
  stat_index = integer,
  stat_event = integer,
  stat_dist = integer,
  stat_explorer = integer,
  stat_explorer_child = integer,
  stat_features = integer
);
type Direction = {
  NORTH,
  WEST,
```

```

    EAST,
    SOUTH,
    ORIGIN
};
type Event = {
    NOEVENT,
    TIMEOUT,
    WAKEUP
};
type Timer = (
    timer_val = integer,
    timer_event = Event
);

```

SEM

Ex. 3. Feature class definitions

```

feature env =
  def private mutable _env_tempobj:simobject =
    null end;
  def mutable blocked = false end;
  def mutable errno = EOK end;
  def private mutable _env_vd = 0.0 end;
  def private mutable _env_vi = 0 end;
  def private mutable _env_i = 0 end;
  def private mutable _env_j = 0 end;
  def private mutable _env_mat:integer list list =
    {} end;
  def private mutable _env_start_time = 0.0 end;
  def private mutable _env_in_await = false end;
  def private mutable _env_vec:integer list =
    {} end;

  def AbsI = fun v:integer ->
    [
      _env_vd := Abs((v::double));
      _env_vi := (_env_vd::integer)
    ]; _env_vi
  end;

  def AwaitDelay = fun tmo:double ->
    if _env_in_await then
      [

```

```

        if ((GetTime()-_env_start_time)>tmo) then
        [
            _env_in_await := false;
            blocked := false
        ]
    ]
else
    [
        _env_in_await := true;
        _env_start_time := GetTime();
        blocked := true
    ];
end;

--
-- CREAT AGENT OF CLASS agentclass
--
def CreateAgent = fun x:integer,y:integer,
                    agentclass:objinstance ->
    CreateObjectAndRemember(agentclass,
        (fun obj:simobject ->
            [
                BeamTo(GetSpatialInfo(obj),
                    CreatePos((x::double)*10.0+5.0,
                        (y::double)*10.0+5.0))
            ]));)
end;

--
-- CREATE AGENT OF CLASS agentclass AND
-- RETURN SIMOBJECT
--
def CreateAgentAndReturn =
    fun x:integer,y:integer,
        agentclass:objinstance ->
    CreateObjectAndRemember(agentclass,
        (fun obj:simobject ->
            [
                _env_tempobj := obj;
                BeamTo(GetSpatialInfo(obj),
                    CreatePos((x::double)*10.0+5.0,
                        (y::double)*10.0+5.0))
            ])); _env_tempobj

```

```

end;
--
-- GET (ONE) AGENT OF SPECIFIED CLASS AT
-- CURRENT POSITION
--
def GetAgent = fun thisclass:objclass ->
  GetFirst(Select(
    (fun obj:simobject ->
      Distance(GetSpatialInfo(obj),
        GetSpatialInfo(GetCurrentSimObject()))
        = -1),
    GetAllObjectsOfType(thisclass,true,true)
  ))
end;
--
-- GET MATRIX ELEMENT mat(i,j) WITH
-- col=0,...,cols-1,row=0,...,rows-1
--
def GetMatrixI = fun mat:integer list list,
  row:integer,col:integer ->
  GetNth(col,GetNth(row,mat))
end;
--
-- COMPUTE MAT1-MAT2 (matrix element subtraction)
--
def MatrixSub = fun mat1:integer list list,
  mat2:integer list list ->
  [
    _env_mat := {};
    _env_j := 0;
    ForElements((fun row:integer list ->
      [
        _env_mat ::+ VectorSub(row,
          GetNth(_env_j,mat2));
        _env_j := _env_j + 1
      ];
    ),mat1)
  ]; _env_mat
end;

def VectorSub = fun vec1:integer list,
  vec2:integer list ->

```

```

[
  _env_vec := {};
  _env_i := 0;
  ForElements((fun col:integer ->
    [
      _env_vec ::+ (col-vec2.[_env_i]);
      _env_i := _env_i + 1
    ]);
    ),vec1)
]; _env_vec
end;

--
-- SET MATRIX ELEMENT mat(i,j) WITH
-- i=0,...,cols-1,j=0,...,rows-1
--
def SetMatrixI =
  fun mat:integer list list,
    row:integer,col:integer,v:integer ->
  [
    _env_vec := GetNth(row,mat);
    SetNth(col,v,_env_vec);
    SetNth(row,_env_vec,mat)
  ];
end;

--
-- I AM THIS SIM OBJECT
--
def Self = fun () -> GetCurrentSimObject()
end;

--
-- CHANGE COLOR
--
def SetColor = fun thisobj:simobject,r,g,b ->
  ChangeShapeColor(GetSpatialInfo(thisobj),
    CreateColor(r,g,b))
end;
end;

```


Ex. 4. Agent class definitions

```
agent sampling (color:orange,
  shape:circle,fill:true,size:{2.0,2.0},
  center:{3.0,1.0}) =
  use os;
  use env;
  use ts_client;

  def mutable adc = 0 end;
  def mutable Pos = Position(0,0) end;
  def mutable self:simobject = null end;
  def mutable myworld:simobject = null end;
  def mutable Parent:simobject = null end;
  def mutable sampled = 0 end;

  activity init =
    [
      self := Self();
      ts_init(Parent);
      myworld := GetWorld()
    ];
  end;
  activity sample =
    [
      adc :=
        GetMatrixI(myworld->myworld->Sensors,
          Pos.Y-1,Pos.X-1);
      out2(ADC,adc);
      sampled := sampled + 1
    ];
  end;
  activity sleep =
    [
      AwaitDelay(10.0)
    ];
  end;

  transition(entry,init);
  transition(init,sample);
  transition(sample,sleep);
  transition(sleep,sample,blocked=false);
```

SEM

end;

VERSION

1.2

Last modified: April 8, 2014

SEM